
BSIM_NMOS_Caps

*
*      IN
*      ___________/\  /\  /\___________________________________
*      |     RG 10K \/  \/                                     |
*      |                                                       |
*      |                                                       |
*      |__________/\  /\  /\_____________                      |
*      |    RDS 10K \/  \/               |                     |
*      |                                 |                     |
*      |                                 |                     |
*      |__________/\  /\  /\__           | 9.1fF               |
*      |    RB  10K \/  \/    |          |                     |
*      |                      |B         |DS                 G | 3.3fF
*      |                      |12fF      |                     |
*     _|__ VIN         MB     |      MDS |            MG       |
*    /_   \          _________|       ___|            ___      |
*   // \   \        |   |_    |      |   |_          |   |_    |
*   \   \_//        |     ||__|      |     ||_       |     ||__|
*    \____/      __/|\____||      __/|\____|| |    _/|\____||
*     _|_       |   |   <-||     |   |   <-|| |   |  |   <-||
*     ///       |   |___|        |   |___|    |   |  |___|
*               |                |            |   |      |
*               |________________|____________|___|______|
*                            _|_
*                            ///
*
.OPTIONS  GMIN=1e-15       METHOD=gear    ABSTOL=1e-15    temp=27
VIN       IN        0      DC     0V      AC    1V
RG        G         IN     10k
RDS       DS        IN     10k
RB        B         IN     10k
MB        B         B      B      0       NMOSC   W=3u     L=1u AD=7p AS=7p PD=10u PS=10u
MDS       DS        0      DS     0       NMOSC   W=3u     L=1u AD=7p AS=7p PD=10u PS=10u
MG        0         G      0      0       NMOSC   W=3u     L=1u AD=7p AS=7p PD=10u PS=10u
.control
run
ac        dec   50  100Meg 10ghz
plot      db(v(g))  db(v(b))    db(v(ds))   ylimit -3 0

let gain =          (db(g))
let f =             mag(frequency[ sortorder( abs(gain+3.0) )[0] ] )
let cg =            1/(6.28e4*f)
echo               "3dB point  f = $&f Hz cg =  $&cg" 

let gain =          (db(b))
let f =             mag(frequency[ sortorder( abs(gain+3.0) )[0] ] )
let cb =            1/(6.28e4*f)
echo               "3dB point  f = $&f Hz cb =  $&cb" 

let gain =          (db(ds))



let f =             mag(frequency[ sortorder( abs(gain+3.0) )[0] ] )
let cds =            1/(6.28e4*f)
echo               "3dB point  f = $&f Hz cds =  $&cds" 

let cbs =           cds/2
let cgb =           cb - cds
let cgs =          (cg - cgb)/2
echo               "cbs = cbd =    $&cbs"
echo               "cgs = cgd =    $&cgs"
echo               "cgb =          $&cgb" 
.endc

.model             NMOSC              NMOS
+ Level=  8        Tnom=27.0
*------------------Process----------------------------------------------- 
+ tox=160e-10      xj=0.25e-06        nch=0.5e+17                        
*------------------V_threshold------------------------------------------- 
+ vth0=0.72        nlx=0.12e-06                                          
*------------------Bulk-------------------------------------------------- 
+ k1=1.04          k2=-1.209E-01
*------------------mobility----------------------------------------------
+ u0=678           ua=8.964e-10                                          
+ ub=1.472e-18     uc=-4.441E-17      vsat=86000                         
*------------------Subthresshold-----------------------------------------
+ nfactor=1.8                                                               
+ cit=-5.0E-04     voff=-7.862E-02                                        
+ eta0=4.441e-16   etab=-2.E-01       dsub=0.7                           
*------------------Hot electrons----------------------------------------- 
+ alpha0=1.61e-05  beta0=36.68                                           
*------------------VAF---------------------------------------------------
+ lint=.12e-06     pclm=.19           pscbe1=3.79e+08   pscbe2=9.4e-05   
+ delta=0.01655    pvag=0.4484                                            
*------------------Bulk_diode-------------------------------------------- 
+ js=5.858e-08                                                              
*------------------Resistance-------------------------------------------- 
+ rsh=70           rdsw=375                                                
+ wr=0.7586        prwb=0             prwg=-4.441E-17                    
*------------------Capacitance-------------------------------------------
+ cj=0.0002424     cjsw=2.73e-10      mj=0.3551           mjsw=0.3873    
+ cgso=9e-13       cgdo=9e-13         cgbo=7e-10                         
+ cdsc=-2.4E-4     cdscd=-1.506E-01   cdscb=-2.219E-04                   
+ pb=0.5614        pbsw=0.8           xpart=0                             
+ dlc=5e-08        dwc=1.5e-07  
*------------------BulkChargeEffect--------------------------------------  
* a0=0.7           a1=0                a2=1                ags=0.05583    
* b0=6.305e-08     b1=6.579e-08        keta=-1.531E-02                    
*------------------ShortChannel------------------------------------------ 
+ dvt0=2.2         dvt1=0.53          dvt2=-1.521E-01      drout=0.76       
+ pdiblcb=.4       pdiblc1=0.00886    pdiblc2=0.00029                      
*------------------NarrowChannel-----------------------------------------
+ w0=2.6e-04       wint=0.16e-06                                          
+ ww=-9.525E-14    wwn=1.0                                               



+ dvt0w=0          dvt1w=5.3e6        dvt2w=-1.E-01                        
+ k3=2.53          k3b=-5             dwg=0                dwb=0  
*------------------Noise-------------------------------------------------
* af=1             kf=1e-28           ef=0.95                            
*------------------Temperature------------------------------------------- 
* pvsat=0          ute=-1.258E+00     kt1=-3.85E-01                      
* kt1l=0           kt2=-3.098E-02     ua1=5.705e-09                      
* ub1=-1.147E-17   uc1=-1.302E-01     at=20380                           
* prt=-3.287E+02   lk1=0              lk2=0                              
* lvsat=0          la0=0              lags=0             lute=0          
* luc=0            
                                                                                          
.end

.end

=====================END_OF_SPICE=======================

Circuit: BSIM_NMOS_Caps

3dB point  f = 4.7863E+09  Hz cg  =  3.3269E-15
3dB point  f = 1.31826E+09 Hz cb  =  1.20793E-14
3dB point  f = 1.7378E+09  Hz cds =  9.16306E-15
cbs = cbd =    4.58153E-15
cgs = cgd =    2.0535E-16
cgb =          2.9162E-15



*
*      IN
*      ___________/\  /\  /\___________________________________
*      |     RG 10K \/  \/                                     |
*      |                                                       |
*      |                                                       |
*      |__________/\  /\  /\_____________                      |
*      |    RDS 10K \/  \/               |                     |
*      |                                 |                     |
*      |                                 |                     |
*      |__________/\  /\  /\__           | 9.6fF               |
*      |    RB  10K \/  \/    |          |                     |
*      |                      |B         |DS                 G | 3.3fF
*      |                      |12fF      |                     |
*     _|__ VIN         MB     |      MDS |            MG       |
*    /_   \          _________|       ___|            ___      |
*   // \   \        |   |_    |      |   |_          |   |_    |
*   \   \_//        |     ||__|      |     ||_       |     ||__|
*    \____/      __/|\____||      __/|\____|| |    _/|\____||
*     _|_       |   |   <-||     |   |   <-|| |   |  |   <-||
*     ///       |   |___|        |   |___|    |   |  |___|
*               |                |            |   |      |
*               |________________|____________|___|______|
*                            _|_
*                            ///
*
*      CG  = CGS + CGD + Cox      CGS = (CG-Cgb)/2



*      CB  = CBS + CBD + Cox      Cgb = CB -CDS
*      CDS = CBS + CBD            CBS = CDS/2
**
**       SPICE MODEL                      ___
**                           __/\  ______|   | Source
**          CGS             |    \/ RS   |___|
**             _____________|__________
**           _|_     _|_        _|_   _|_
**     Gate  ___    / _ \ I_ds  \ /   ___ CBS  6fF<- (Caps for L/W 1u/2.6u)
**      ___   |     \/ \/       _V_    |      
**     |   |__|     /\_/\  |     |_____|     ___   
**     |___|  |___  \___/  V  ___|     |____|   | Bulk   10K10F 100p
**            |   |   |      |  _|_   _|_   |___|
**           _|_  |__/|\__||_|   ^    ___     |__..no..__
**           ___      |2fF||    /_\    |  CBD           _|_
**            |_______|__________|_____|  6fF            ^ 
**           CGD        CGD |             ___           /_\
**                          |__/\  ______|   | Drain    _|_
**                               \/ RD   |___|         \sub/
**                                                      \ /
**                         <Bulk_hidden = gnd!>          V
**   
**                  (G)  
**           (D)   ___|__   (S)        (B)         (Sub)
**           _|_  |______|  _|_        _|_          _|_ 
**    ______|___|_|______|_|___|______|___|________|___|__ _
**     |  |   P+  | <++++|   P+  |  |   N+  | |  |   P+  |   Xj
**     |   \_____/        \_____/    \_____/  |   \_____/  _
**     |                NWELL                 |
**     \______________________________________/ 
**   
**   


